Introduction to Experimental Mathematics

Victor Adamchik

Carnegie Mellon University

Online Mathematical Tools

"One of the greatest ironies of the information technology revolution is that while the computer was conceived and born in the field of pure mathematics <>, until recently this marvelous technology had only a minor impact within the field that gave it birth"

Jon Borwein [1]

21 Century Program [J. Borwein]

Question →

Computer Algebra System →

Online Math. Tools →

Search Engine →

Digital Libraries →

Answer

Encyclopedia of Integer Sequences

http://oeis.org/

■ **Finite Sum**

Find a closed form

$$\sum_{k=0}^{n} \binom{n}{k}^2$$

We compute a few first numbers

```
Table[Sum[Binomial[n, k]^2, {k, 0, n}], {n, 1, 7}]
```

```
{2, 6, 20, 70, 252, 924, 3432}
```

and then search

http://oeis.org/

to make the following conjecture

$$\sum_{k=0}^{n} \binom{n}{k}^2 = \binom{2\,n}{n}$$

■ Finite Sum

Find a closed form ($n \geq 2$)

$$\sum_{k=0}^{n} \left(18\,k^2 - 9\,k\,n + 3\,k - 8\,n - 12\right) \binom{n+4}{3\,k-n}$$

Compute a few first numbers

```
Table[Sum[(18 k^2 - 9 k n + 3 k - 8 n - 12) Binomial[n + 4, 3 k - n], {k, 0, n}], {n, 2, 7}]
```

```
{60, -84, 112, -144, 180, -220}
```

take the absolute value

```
Abs[%]
```

```
{60, 84, 112, 144, 180, 220}
```

and then search the table of sequences

```
Table[4 Binomial[n, 2], {n, 2, 11}]
```

```
{4, 12, 24, 40, 60, 84, 112, 144, 180, 220}
```

We conjecture

$$\sum_{k=0}^{n} \left(18 k^2 - 9 k n + 3 k - 8 n - 12\right) \binom{n+4}{3k-n} = (-1)^n \, 4 \binom{n+4}{2}$$

■ Recurrence Equation

Find a closed form for a_n:

$$a_{n+1} = \frac{a_n + 2}{2 a_n + 3}$$
$$a_0 = 1$$

We compute a few first numbers

```
a[0] = 1;
a[n_] := a[n] = (a[n-1] + 2) / (2*a[n-1] + 3)
```

```
Table[a[n], {n, 0, 6}]
```

```
{1, 3/5, 13/21, 55/89, 233/377, 987/1597, 4181/6765}
```

and then search

http://oeis.org/

for the numerators 1, 3, 13, 55, 233 and the denominators 1, 5, 21, 89, 377 separately. Both searches return the Fibonacci numbers. Therefore, a closed form is given by

$$a_n = \frac{F_{3n+1}}{F_{3n+2}}$$

Once, the form is guessed, it is easy to prove it.

■

■ A curious anomaly

Consider a series for π (the Gregory series)

$$\pi = 4 \sum_{k=1}^{\infty} \frac{(-1)^{k-1}}{2k-1}$$

A series truncated to 5 million digits

3.141592**4**53589793238462**4**6433832795027**7**841971693993**873**0582097494**182**230781640
...

Here are actual digits of π

3.141592**6**5358979323846**2**643383279502**8**841971693993**751**0582097494**459**230781640 ...

This behavior was first observed by J. R. North in 1988

Consider differences:

```
{6 - 4, 4 - 2, 8 - 7, 873 - 751, 459 - 182}
```

```
{2, 2, 1, 122, 277}
```

We slightly modified the sequence

```
{6 - 4, 4 - 2, 88 - 78, 873 - 751, 4592 - 1822}
```

Divide this by 2 and enter `"1, 1, 5, 61, 1385"` into the Online Encyclopedia of Integer Sequences. We find that this is a sequence for the Euler numbers

```
Table[EulerE[2 k], {k, 0, 5}]
```

```
{1, -1, 5, -61, 1385, -50 521}
```

that are defined by

$$\sec(x) = \sum_{k=0}^{\infty} \frac{(-1)^k \, E_{2k}}{(2k)!} x^{2k}$$

What did we discover?? An asymptotic expansion

$$\pi - 4 \sum_{k=1}^{n/2} \frac{(-1)^{k-1}}{2k-1} \approx \sum_{k=0}^{\infty} \frac{E_{2k}}{4^k \, n^{2k+1}}$$

The Inverse Symbolic Calculator

What is Inverse Symbolic Computation? In short, "reverse engineering" of real numbers. Given a nunber or sequence of numbers find where they come from.

https://isc.carma.newcastle.edu.au/

http://oldweb.cecm.sfu.ca/projects/ISC/ISCmain.html

■ Problem from the previous lecture

Use Gröbner bases to find the minimal distance between the roots of

$$z^3 + c^2 z + 1 = 0$$

for real $c > 0$. Here is a graphic of the numerical calculated minimum



We find c numerically

```
minDistance[c_] :=
 Module[{roots, p, dist},
  roots = x /. NSolve[x^3 + c^2 x + 1 == 0, x, WorkingPrecision → 50];
  p = Partition[roots, 2, 1, 1];
  dist = Abs[#[[1]] - #[[2]] & /@ p];
  Min[dist]
 ]
```

```
FindMinimum[minDistance[c], {c, 1}, WorkingPrecision → 40]
```

```
{1.543081844217052283611875212073110387083,
 {c → 1.091123635971721403554616991249479878576}}
```

Let us search the Inverse Symbolic Calculator at

http://oldweb.cecm.sfu.ca/projects/ISC/ISCmain.html

Feeding in 1.09112363 yields

$$\frac{\sqrt{3}}{2^{2/3}}$$

- **Integral Equation**
- **Trigonometry**

What is the algebraic value of

$$-\sqrt[3]{\cos\left(\frac{2}{7}\pi\right)} + \sqrt[3]{-\cos\left(\frac{4}{7}\pi\right)} + \sqrt[3]{-\cos\left(\frac{6}{7}\pi\right)}$$

Compute this numerically

$$N\left[-\sqrt[3]{\cos\left(\frac{2}{7}\pi\right)}+\sqrt[3]{-\cos\left(\frac{4}{7}\pi\right)}+\sqrt[3]{-\cos\left(\frac{6}{7}\pi\right)},75\right]$$

```
0.717515079649939935120950559177986112108457601155250572183302830027981⋱
 465015
```

Cube it and then use "Integer Relation Algorithms" option

```
0.717515079649939935^3
```

```
0.36939677415858365
```

You will get back the following information

K satisfies the following polynomial:

$$-32+75x+30x^2+4x^3$$

In[22]:=

```
a = 0.36939677415858365;
B := {{1, 0, 0, 0, c },
    {0, 1, 0, 0, c a},
    {0, 0, 1, 0, c a^2},
    {0, 0, 0, 1, c a^3}};
c = 10^15;
Round[N[B, 30]];
LatticeReduce[%][[1]];
N[%]
```

Out[27]=

```
{-32., 75., 30., 4., 36.}
```

```
Solve[-32 + 75 x + 30 x^2 + 4 x^3 == 0, x]
```

$$\left\{\left\{x\to-\frac{5}{2}+\frac{3\times 7^{1/3}}{2}\right\},\left\{x\to-\frac{5}{2}-\frac{3}{4}7^{1/3}\left(1-i\sqrt{3}\right)\right\},\left\{x\to-\frac{5}{2}-\frac{3}{4}7^{1/3}\left(1+i\sqrt{3}\right)\right\}\right\}$$

It follows that

$$-\sqrt[3]{\cos\left(\frac{2}{7}\pi\right)}+\sqrt[3]{-\cos\left(\frac{4}{7}\pi\right)}+\sqrt[3]{-\cos\left(\frac{6}{7}\pi\right)}=\sqrt[3]{-\frac{5}{2}+\frac{3\sqrt[3]{7}}{2}}$$

■ Simple Integral

■ Definite Integral

Consider the integral

$$\int_0^{\frac{\pi}{4}} \frac{t^2}{\sin^2 t}\, dt$$

and ask what is its analytic value? We compute this to hundred digits

```
NIntegrate[t^2 / Sin[t]^2, {t, 0, Pi / 4}, WorkingPrecision → 30]
```

```
0.843511841685034634002620052000
```

and then send it to the Inverse Symbolic Calculator (choose Integer Relation Algorithm as an option). We get back

K satisfies the following Z-linear combination :

- 16 K - Pi**2 + 16 Catalan + 4 Pi*log(2)

Thus we conjecture

$$\int_0^{\frac{\pi}{4}} \frac{t^2}{\sin^2 t}\, dt = \frac{\pi \log 2}{4} - \frac{\pi^2}{16} + C$$

where C is Catalan's contant in *Mathematica*.

```
In[48]:= v = 0.843511841685034634002620051999; B := {{1, 0, 0, 0, c v},
           {0, 1, 0, 0, c Pi^2},
           {0, 0, 1, 0, c Catalan},
           {0, 0, 0, 1, c Pi Log[2]}};
         c = 10^15;
         Round[N[B, 30]];
         LatticeReduce[%][[1]];
         N[%]
```

```
Out[52]= {-16., -1., 16., 4., -7.}
```

```
In[53]:= -16 v - Pi^2 + 16 Catalan + 4 Pi Log[2]
```

```
Out[53]= 0. × 10^-29
```

Falsy Patterns

■ Numeric Fraud

Consider the following series

$$\sum_{k=0}^{\infty} \frac{\lfloor k \tanh(\pi) \rfloor}{10^k}$$

and let us compute it numerically.

```
NSum[Floor[k * Tanh[Pi]]/10^k, {k, 0, Infinity}]
```

```
0.0123457
```

This suggests

$$\sum_{k=0}^{\infty} \frac{\lfloor k \tanh(\pi) \rfloor}{10^k} = \frac{1}{81}$$

Let us recompute the series with higher precision

100 digits

```
1/81 - Sum[N[Floor[k * Tanh[Pi]]/10^k, 100], {k, 0, 2000}]
```

```
0. × 10^-102
```

300 digits

```
1/81 - Sum[N[Floor[k * Tanh[Pi]]/10^k, 300], {k, 0, 2000}]
```

```
1.11111111111111111111111111111111 × 10^-269
```

As you see the sum is $\frac{1}{81}$ up to 268 digits!!

■ Symbolic Fraud

Consider the following class of integral

$$\int_0^{\infty} \prod_{k=1}^{n} \operatorname{sinc}\left(\frac{x}{2\,k-1}\right) dx$$

$$\operatorname{sinc}(x) = \frac{\sin x}{x}$$

What is its closed form? Here are few particular cases

```
Integrate[Sinc[x], {x, 0, Infinity}]
```

$\frac{\pi}{2}$

```
Integrate[Sinc[x] Sinc[x/3], {x, 0, Infinity}]
```

$\frac{\pi}{2}$

```
Integrate[Sinc[x] Sinc[x/3] Sinc[x/5], {x, 0, Infinity}]
```

$\frac{\pi}{2}$

```
Integrate[Sinc[x] Sinc[x/3] Sinc[x/5] Sinc[x/7], {x, 0, Infinity}]
```

$\frac{\pi}{2}$

We are ready to make a conjecture!!

$$\int_0^\infty \prod_{k=1}^{n} \operatorname{sinc}\left(\frac{x}{2k-1}\right) dx = \frac{\pi}{2}$$

Unfortunately,

```
Integrate[Sinc[x] Sinc[x/3] Sinc[x/5] Sinc[x/7] Sinc[x/9] Sinc[x/11]
  Sinc[x/13] Sinc[x/15], {x, 0, Infinity}]
```

$$\frac{467\,807\,924\,713\,440\,738\,696\,537\,864\,469\,\pi}{935\,615\,849\,440\,640\,907\,310\,521\,750\,000}$$

```
N[% / Pi, 20]
```

```
0.49999999999264685932
```

References

[1] J. Borwein and D, Bailey, *Mathematics by Experiment: Plausible Reasoning in the 21st Century*, AK Peters Ltd, 2003.